Fairlane Wheel Centrifugal Stresses

The Team 100 shooter uses Fairlane neoprene rubber wheels, McMaster-Carr part number 2476K37. There is a question whether the wheels can be spun at the speeds necessary to launch power cells (foam balls) in the 2020 Infinite Recharge FRC game without over-stressing the wheels. Accordingly, we have performed a axisymmetric stress analysis to calculate the radial and hoop stresses and the radial displacements in the rubber and steel. The wheels are 4-inch-diameter by 2-inch wide with a tubular steel core having 1.25-inch ID and 1/16-inch wall thickness. The Durometer is 60A. In the analysis, Material 1 is designated by subscript 1 and represents the steel core. Material 2 is the rubber.

Roller

4" Diameter, 1-15/16" Width



Durometer (Hardness Rating)

✓ ● 60A (Medium) Black

Each

ADD TO ORDER

Delivers Friday 9-11 am
$20.84 Each
2476K37



Guide Roller Type	Drive
Roller Style	Shaft Mount
Roller Profile	Flat
Roller Material	Neoprene
Hub Material	Steel
Roller	
Diameter	4"
Width	1 15/16"
For Shaft Diameter	1 1/4"
Shaft Mount Type	Press Fit
Temperature Range	-40° to 200° F
Durometer (Hardness Rating)	60A (Medium) Black
RoHS	RoHS 3 (2015/863/EU) compliant

Made of neoprene rubber, these rollers resist oil, flames, gasoline, and weather. Mount them directly onto your shaft or build a custom roller by adding your own hub.

✓ **Durometer (Hardness Rating): 60A (Medium) Black**

Neoprene (chloroprene) Durometer 60A rubber properties are taken from the MatWeb page http://www.matweb.com/search/DataSheet.aspx?MatGUID=c680960fc38f488a991e1e04504092b4. The relevant properties for the stress analysis are the modulus of elasticity (Young's modulus) and density (specific gravity).

Seals Eastern 7282 Neoprene / Chloroprene

Categories:	Polymer; Thermoset; Rubber or Thermoset Elastomer (TSE)
Material Notes:	UL-flame resistant, non-blooming neoprene Information provided by Seals Eastern, Inc.
Key Words:	Polychloroprene Rubber (CR)
Vendors:	No vendors are listed for this material. Please click here if you are a supplier and would like information on how to add your listing to this material.

Physical Properties	Metric	English	Comments
Specific Gravity	1.52 g/cc	1.52 g/cc	ASTM D297, 15

Mechanical Properties	Metric	English	Comments
Hardness, Shore A	>= 58	>= 58	ASTM D2240, Type A
	62	62	AVG; ASTM D2240, Type A
	73 @Temperature 100 °C, Time 605000 sec	73 @Temperature 212 °F, Time 168 hour	After Dry Heat Aging; ASTM D865
Tensile Strength at Break	>= 11.88 MPa	>= 1723 psi	ASTM D412 Method A
	14.84 MPa	2152 psi	AVG; ASTM D412 Method A
	12.47 MPa @Temperature 100 °C, Time 605000 sec	1808 psi @Temperature 212 °F, Time 168 hour	After Dry Heat Aging; ASTM D865
Tensile Strength, Yield	1.00 MPa	145 psi	AVG Stress at Elongation M25; ASTM D412 Method A
	>= 1.03 MPa	>= 149 psi	Stress at Elongation M50; ASTM D412 Method A
	1.49 MPa	216 psi	AVG Stress at Elongation M50; ASTM D412 Method A
	>= 1.74 MPa	>= 252 psi	Stress at Elongation M100; ASTM D412 Method A
	2.62 MPa	380 psi	AVG Stress at Elongation M100; ASTM D412 Method A
Elongation at Break	>= 287 %	>= 287 %	ASTM D412 Method A
	362 %	362 %	AVG; ASTM D412 Method A
	166.5 % @Temperature 100 °C, Time 605000 sec	166.5 % @Temperature 212 °F, Time 168 hour	After Dry Heat Aging; ASTM D865
Modulus of Elasticity	0.00614 GPa	0.890 ksi	ASTM D412 Method A
Shear Modulus	0.00205 GPa	0.297 ksi	ASTM D412 Method A
Secant Modulus	0.00539 GPa	0.782 ksi	ASTM D412 Method A
Tear Strength Test	198	198	(psi); ASTM D624, Die B
Compression Set	17 % @Temperature 100 °C, Time 252000 sec	17 % @Temperature 212 °F, Time 70.0 hour	ASTM D395, Test B

Thermal Properties	Metric	English	Comments
Glass Transition Temp, Tg	-40.0 °C	-40.0 °F	ASTM D1329

Descriptive Properties	
Strain Energy / Unit Volume at 20% Elongation (psi)	16

Some of the values displayed above may have been converted from their original units and/or rounded in order to display the information in a consistent format. Users requiring more precise data for scientific or engineering calculations can click on the property value to see the original value as well as raw conversions to equivalent units. We advise that you only use the original value or one of its raw conversions in your calculations to minimize rounding error. We also ask that you refer to MatWeb's terms of use regarding this information. Click here to view all the property values for this datasheet as they were originally entered into MatWeb.

The “Tensile Strength, Yield” values given above are not yield stresses in the traditional sense of the onset of permanent (plastic) deformation, but rather stresses at 25%, 50%, and 100% elongations as means for characterizing the nonlinear stress-strain curve at large strains. As will be shown, predicted strains at 10000 RPM are around 27% maximum. See https://moldeddimensions.com/blog/defining-tensile-strength-elongation-and-modulus-for-rubber-and-cast-polyurethane-materials/.

The analysis solves for the stresses and radial displacement as a function of the radius. The analysis is well known, and is typically used to evaluate spinning rotors, thick-walled pressure vessels, and interference fits.



Parameters

Parameters used in the analysis are:

a = inside radius = 0.625 in = 15.9 mm

b = radius at interface between steel and rubber = 0.6875 in = 17.5 mm

c = outside radius = 2.0 in = 51 mm

σrr, ϵrr = radial stress and strain components

$\sigma\theta\theta$, $\epsilon\theta\theta$ = hoop stress and strain components

σzz, ϵzz = axial stress and strain components (σzz = 0 in the assumed case of plane stress applicable to a relatively thin disk, and ϵzz is not evaluated, although it could be.)

ν1 = Poisson's ratio of steel = 0.3

ν2 = Poisson's ratio of rubber = 0.5 (I.e., in usual fashion, rubber is assumed incompressible. That means the shear modulus is significantly less than the bulk modulus)

e1 = Young's modulus of steel = 209 GPa

e2 = Young's modulus of 60A Durometer rubber = 6.14 MPa

ρ1 = density of steel = 8000 kg/m^3

ρ2 = density of polychloroprene rubber = 1520 kg/m^3

Equations

The equilibrium equations for the two materials in terms of stress are:

In[1]:=
```
deqs = {
   σrr1'[r] + 1/r (σrr1[r] - σθθ1[r]) + ρ1 ω^2 r == 0,
   σrr2'[r] + 1/r (σrr2[r] - σθθ2[r]) + ρ2 ω^2 r == 0
  };
```

(See R. Budynas, *Advanced Strength and Applied Stress Analysis* (1977, McGraw-Hill, p. 139.)

The strain-displacement compatibility equations for the two materials are:

Compatibility:

$$\boxed{\frac{\partial^2 \epsilon_\theta}{\partial r^2} + \frac{1}{r^2}\frac{\partial^2 \epsilon_r}{\partial \theta^2} + \frac{2}{r}\frac{\partial \epsilon_\theta}{\partial r} - \frac{1}{r}\frac{\partial \epsilon_r}{\partial r} = \frac{1}{r}\frac{\partial^2 \gamma_{r\theta}}{\partial r \partial \theta} + \frac{1}{r^2}}$$

(See http://www.ecs.umass.edu/~arwade/courses/str-mech/polar.pdf)

The compatibility equations are second order differential equations in terms of strain or stress, but can be reduced to first order equations.

In[2]:=
```
compateqs = {
   r (D[ e1 ϵθθ1[r], {r, 2}] + 2/r D[e1 ϵθθ1[r], r] - 1/r D[e1 ϵrr1[r], r]) == 0,
   r (D[e2 ϵθθ2[r], {r, 2}] + 2/r D[e2 ϵθθ2[r], r] - 1/r D[e2 ϵrr2[r], r]) == 0
  };
```

The roller is relatively short, and the end faces are traction free, so we will assume plane stress conditions. In that case, the relationships between strains and stresses are:

Hooke's law:
Plane stress

$$\epsilon_r = (\sigma_r - \nu\sigma_\theta)/E, \qquad \epsilon_\theta = (\sigma_\theta - \nu\sigma_r)/E, \qquad \gamma_{r\theta} = \tau_{r\theta}/G$$

(See http://www.ecs.umass.edu/~arwade/courses/str-mech/polar.pdf)
Due to axisymmetry, there is no shear stress.

The stress-strain relations (constitute equations) are:

In[3]:=
```
ϵrr1[r_] := 1/e1 (σrr1[r] - ν1 σθθ1[r])
```

In[4]:=
```
ϵrr2[r_] := 1/e2 (σrr2[r] - ν2 σθθ2[r])
```

In[5]:=
```
ϵθθ1[r_] := 1/e1 (σθθ1[r] - ν1 σrr1[r])
```

In[6]:=
```
ϵθθ2[r_] := 1/e2 (σθθ2[r] - ν2 σrr2[r])
```

Substituting strains in terms of stresses in the compatibility equations leads to the following second order equations:

```
In[7]:= compateqs // Simplify

Out[7]= {(2 + ν1) σθθ1′[r] + r (-ν1 σrr1″[r] + σθθ1″[r]) == (1 + 2 ν1) σrr1′[r],
  (2 + ν2) σθθ2′[r] + r (-ν2 σrr2″[r] + σθθ2″[r]) == (1 + 2 ν2) σrr2′[r]}
```

Integrating by parts and recognizing that $\int r\,\sigma''[r]\,dr == r\,\sigma'[r] - \sigma[r]$ to eliminate the second derivatives yields first order differential equations.

```
In[8]:= compat = {
    (1 + 2 ν1) σrr1[r] - (2 + ν1) σθθ1[r] ==
     -ν1 (r D[σrr1[r], r] - σrr1[r]) + (r D[σθθ1[r], r] - σθθ1[r]),
    (1 + 2 ν2) σrr2[r] - (2 + ν2) σθθ2[r] ==
     -ν2 (r D[σrr2[r], r] - σrr2[r]) + (r D[σθθ2[r], r] - σθθ2[r])
   }

Out[8]= {(1 + 2 ν1) σrr1[r] - (2 + ν1) σθθ1[r] == -σθθ1[r] - ν1 (-σrr1[r] + r σrr1′[r]) + r σθθ1′[r],
  (1 + 2 ν2) σrr2[r] - (2 + ν2) σθθ2[r] == -σθθ2[r] - ν2 (-σrr2[r] + r σrr2′[r]) + r σθθ2′[r]}
```

Boundary Conditions

Boundary conditions are such that the bore and OD are traction-free (i.e., radial stress is zero there), and the radial stress is continuous across the rubber-steel interface.

```
In[9]:= bcs1 = {
    σrr1[a] == 0,
    σrr2[c] == 0,
    σrr1[b] == σrr2[b]
   };
```

If there is an interference fit, then the inward displacement of the core at its OD and the outward displacement of the rubber at its ID must add up to the initial radial interference fit when the parts are separate. In the case of the Fairlane wheels, we assume there is no significant interference fit. We use the fact that $\epsilon\theta\theta = u/r$ to obtain the displacement condition for $u[b]$ at the interface between steel and rubber in terms of the stresses via the $\epsilon\theta\theta$ strains and the constitutive equations.

```
In[10]:= bcs2 = (
     b ϵθθ2[b] - b ϵθθ1[b] == δ
    );

In[11]:= bcs = Flatten[{bcs1, bcs2}];
```

Solution

Solve the differential equations symbolically subject to the compatibility equations and boundary conditions.

```
In[12]:= ansr = Flatten[DSolve[Flatten[{deqs, compat, bcs}], {σrr1, σθθ1, σrr2, σθθ2}, r]];
```

Use the fact that $\epsilon\theta\theta = u/r$ to solve for u in terms of the stresses.

In[13]:=
```
u1[p_] := p ϵθθ1[p]
```

In[14]:=
```
u2[p_] := p ϵθθ2[p]
```

Set up parameters for calculating the specific wheel at issue and for plotting.

In[15]:=
```
paramsPstress = {e1 → 200. × 10^9, e2 → 6.14 × 10^6, ν1 → 0.29, ν2 → 0.5, δ → 0.000,
   a → 0.0159, b → 0.0175, c → 0.051, ρ1 → 8000., ρ2 → 1520., ω → 1050};
```

Checks and Verifications

Check that the radial stress components are continuous, that is, their difference is zero, across the interface at $r = b$ under the plane stress assumption.

In[16]:=
```
σrr1[b] - σrr2[b] /. ansr /. paramsPstress
```

Out[16]=
```
0.
```

Note that the hoop ($\sigma\theta\theta$) stresses are discontinuous across the interface.

In[17]:=
```
σθθ1[b] - σθθ2[b] /. ansr /. paramsPstress
```

Out[17]=
```
2.41721 × 10^7
```

Check that the displacements are continuous across the interface between the rubber and steel under the plane stress assumption.

In[18]:=
```
u2[b] - u1[b] - δ /. ansr /. paramsPstress
```

Out[18]=
```
2.50425 × 10^-18
```

Verify that small strain theory is relevant:

In[19]:=
```
{ϵrr2[b], ϵrr2[c], ϵθθ2[c]} /. ansr /. paramsPstress
```

Out[19]=
```
{0.267983, -0.0332404, 0.0664809}
```

Note that the radial strain near the interface is moderately large, but still less than one-tenth the strain at break listed above for 60A Durometer chloroprene rubber.

Stress Distributions

In[20]:=
```
g1 = Plot[Evaluate[{σrr1[r], σθθ1[r]} /. ansr /. paramsPstress], {r, 0.0159, 0.0175},
   PlotRange → {{0.015, 0.051}, {-0 × 10^7, 3 × 10^7}}, Frame → True, FrameLabel →
    {"RADIUS (m)", "STRESS (Pa)", "Plane Stress"}, PlotLegends → {"σrr", "σθθ"}];
```

In[21]:=
```
g1a = Plot[Evaluate[{σrr1[r], σθθ1[r]} /. ansr /. paramsPstress], {r, 0.0159, 0.0175},
   PlotRange → {{0.015, 0.051}, {-0 × 10^7, 2.5 × 10^6}}, Frame → True, FrameLabel →
    {"RADIUS (m)", "STRESS (Pa)", "Plane Stress"}, PlotLegends → {"σrr", "σθθ"}];
```

In[22]:=
```
g2 = Plot[Evaluate[{σrr2[r], σθθ2[r]} /. ansr /. paramsPstress],
    {r, 0.0175, 0.051}, PlotRange → {{0.015, 0.051}, {-0 × 10^7, 3 × 10^7}},
    Frame → True, FrameLabel → {"RADIUS (m)", "STRESS (Pa)", "Plane Stress"},
    PlotStyle → {{Blue, Thickness[0.015]}, {Red, Thick}}];
```

In[23]:=
```
g2a = Plot[Evaluate[{σrr2[r], σθθ2[r]} /. ansr /. paramsPstress],
    {r, 0.0175, 0.051}, PlotRange → {{0.015, 0.051}, {-0 × 10^7, 2.5 × 10^6}},
    Frame → True, FrameLabel → {"RADIUS (m)", "STRESS (Pa)", "Plane Stress"},
    PlotStyle → {{Blue, Thickness[0.015]}, {Red, Thick}}];
```

The hoop stress is the steel is the highest stress by far.

In[24]:=
```
Show[g1, g2]
```

Out[24]=



Zooming in on the rubber stress distribution.

In[25]:=
```
Show[g1a, g2a]
```

Out[25]=



Displacement Distributions

The steel expands (displaces radially) negligibly compared to the rubber.

In[26]:=
```
g3 = Plot[Evaluate[{u1[r]} /. ansr /. paramsPstress],
   {r, 0.0159, 0.0175}, PlotRange → {{0.015, 0.051}, {0, 0.005}}, Frame → True,
   FrameLabel → {"RADIUS (m)", "RADIAL DISPLACEMENT (m)", "Plane Stress"},
   PlotLegends → {"Steel Core"}, PlotStyle → {Thickness[0.02], Red}];
```

In[27]:=
```
g4 = Plot[Evaluate[{u2[r]} /. ansr /. paramsPstress],
   {r, 0.0175, 0.051}, PlotRange → {{0.015, 0.051}, {0, 0.005}}, Frame → True,
   FrameLabel → {"RADIUS (m)", "RADIAL DISPLACEMENT (m)", "Plane Stress"},
   PlotLegends → {"Rubber Roller"}, PlotStyle → Black];
```

In[28]:=
```
Show[g3, g4]
```

Out[28]=



The Effect of Speed

Stresses and displacements vary as the square of the speed.

Remove the speed from the set of parameters to enable plotting versus speed.

In[29]:=
```
paramsPstress2 = {e1 → 200. × 10^9, e2 → 6.14 × 10^6, ν1 → 0.29, ν2 → 0.5,
   δ → 0.000, a → 0.0159, b → 0.0175, c → 0.051, ρ1 → 8000., ρ2 → 1520.};
```

Stresses

The radial stress at the interface grows as the square of the speed.

In[30]:=
```
g5 = ParametricPlot[{60/(2 π) ω, Evaluate[σrr2[b] /. ansr /. paramsPstress2]},
  {ω, 0, 1000}, PlotRange → {{0, 10 000}, {0, 2.5 × 10^6}},
  Frame → True, AspectRatio → 1/GoldenRatio, FrameLabel →
   {"SPEED (RPM)", "INTERFACE STRESS (Pa)", "Plane Stress"}, PlotStyle → Black]
```

Out[30]=



In[31]:=
```
UnitConvert[1. × 10^6 Pa, lbf/in^2]
```

Out[31]= $145.038\ \mathrm{lbf/in^2}$

The maximum radial stress occurs at the interface, and is approximately 300 psi at 9,500 RPM.

In[32]:=
```
g5 = ParametricPlot[{60/(2 π) ω, Evaluate[145.038/10^6 σrr2[b] /. ansr /. paramsPstress2]},
  {ω, 0, 1000}, PlotRange → {{0, 10 000}, {0, 350}},
  Frame → True, AspectRatio → 1/GoldenRatio, FrameLabel →
   {"SPEED (RPM)", "INTERFACE STRESS (psi)", "Plane Stress"}, PlotStyle → Black]
```

Out[32]=



Evaluate the hoop stress in the steel core, which is highest at the inside radius *a*:

In[33]:=
```
ParametricPlot[{60/(2 π) ω, Evaluate[145.038/10^6 σθθ1[a] /. ansr /. paramsPstress2]},
 {ω, 0, 1050}, PlotRange → {{0, 10 000}, {0, 5000}},
 Frame → True, AspectRatio → 1/GoldenRatio, FrameLabel →
  {"SPEED (RPM)", "CORE HOOP STRESS (psi)", "Plane Stress"}, PlotStyle → Black]
```

Out[33]=



Displacements

The roller grows approximately 6 mm in diameter or 0.25 in at 9,500 RPM.

In[34]:=
```
g4b = ParametricPlot[{60/(2 π) ω, Evaluate[u2[c] /. ansr /. paramsPstress2]},
  {ω, 0, 1000}, PlotRange → {{0, 10 000}, {-0.0001, 0.005}},
  Frame → True, AspectRatio → 1/GoldenRatio, FrameLabel →
   {"SPEED (RPM)", "RADIAL DISPLACEMENT (m)", "Plane Stress"}, PlotStyle → Red]
```

Out[34]=



In[35]:=
```
g4c = ParametricPlot[{60/(2 π) ω, Evaluate[1/(25.4 × 10^-3) u2[c] /. ansr /. paramsPstress2]},
  {ω, 0, 1000}, PlotRange → {{0, 10 000}, {-0.0001, 0.2}},
  Frame → True, AspectRatio → 1/GoldenRatio, FrameLabel →
   {"SPEED (RPM)", "RADIAL DISPLACEMENT (in)", "Plane Stress"}, PlotStyle → Red]
```

Out[35]=



Conclusions

Radial stresses at the interface are maximum 300 psi at 9,500 RPM as compared to a tensile strength at break of >1723 psi for the chloroprene referenced above. At 9,500 RPM, radial strains in the rubber at the interface are 27% as compared to an elongation at break of >287%. Hoop stresses in the steel are maximum 4,000 psi at 9,500 RPM as compared to 36,000 psi for garden-variety ASTM A36 steel.

This analysis does not at all support the wrapping the specific 60A-Durometer Fairlane rollers with stainless steel "safety" wire. The analysis indicates that stresses, strains, and displacements would not exceed values that the rubber would be expected to withstand, with considerable margin, provided the mechanism providing the rotation itself was robust and controlled.

Limitations

This analysis assumes that the Seals Eastern neoprene parameters are applicable to Fairlance 60A rollers. We would not expect significant differences. This analysis also assumes that the rollers are free of defects, including manufacturing defects or holes drilled for "safety" wires. This analysis is only applicable to the specific geometry, materials, and use conditions modelled. Any use of this analysis is at the users sole risk.